

Práctica 2

Con esta práctica, el alumnado debe adquirir soltura en las operaciones trabajadas en la práctica anterior, especialmente en la utilización de referencias absolutas y relativas y el copiado de fórmulas. Además se incluyen algunas novedades en la utilización de funciones (el condicional) y en la forma de rellenar filas o columnas con una serie.

Continuaremos con las funciones elementales: edición, desplazamiento por la hoja, guardar, modificar, reemplazar, copiar, mover, pegar.

Se pretende que el usuario o usuaria revise los conocimientos adquiridos en la práctica anterior y amplíe sus conocimientos sobre el uso de fórmulas y funciones. Continuaremos con el estudio de algunas características de las celdas.

Cuadro de amortización de un préstamo

Se trata de obtener el cuadro de amortización de un préstamo, aplicación que puede servir de apoyo a la comprensión de los conceptos de interés compuesto y anualidades de amortización constantes.

La estructura de la hoja resultante debe ser válida para cualesquiera capitales, tipos de interés y períodos de amortización no superior a 20 años (aunque se puede extender este período hasta alcanzar el número de años deseado, dentro de las limitaciones del programa).

Se deben obtener la anualidad de amortización, el capital amortizado en cada período, el desglose de la anualidad en capital e intereses en cada momento, el capital a solicitar en función de un período de amortización, un tipo de interés y una anualidad determinados y el período de amortización correspondiente a un capital, un tipo de interés y una anualidad definidos previamente.

Descripción de la hoja de cálculo

Empezamos colocando el nombre de la hoja, **Cuadro de amortización de un préstamo**. No nos preocupamos, de momento, del ancho de las celdas ni de la alineación.

A continuación, asignamos las celdas que contendrán los datos iniciales y aquellos que la hoja debe calcular, de acuerdo con la tabla siguiente:

B4	Capital	B5	Interés
B6	Años	E5	Anualidad
B8	Total de intereses	A10	Año
B10	Capital inicial	C10	Principal
D10	Intereses	E10	Capital final

Conviene fijarse en la modificación del ancho estándar de las columnas **A**, **B** y **E**. Para cambiar el ancho de una columna, colocar el cursor sobre la línea que separa una columna de otra (en la línea de nombres de columna) y cuando el cursor cambia de forma, desplazar la línea de división a derecha (aumentar ancho) o izquierda (disminuir). O bien, con el cursor sobre una celda cualquiera de la columna, elegir Ancho... (o Autoajustar a la selección) en el comando **Columna** ► del menú **Formato**.

Ahora tenemos que teclear las cantidades iniciales y las fórmulas adecuadas para calcular los valores deseados.

El **capital** (préstamo solicitado) se colocará en **C4**, el **interés** en **C5** y el número de **años** (período de amortización) en **C6**. De momento introducimos un cero en estas celdas.

	A	B	C	D	E	F
1			CUADRO DE AMORTIZACION DE UN PRESTAMO			
2						
3						
4		Capital				
5		Interés			Anualidad ..	
6		Años				
7						
8		Total intereses				
9						
10	Año	Capital inicial	Principal	Intereses	Capital final	

La celda **F5** contendrá la cuantía de la anualidad correspondiente de acuerdo con los datos iniciales, es decir, $A = \frac{C \cdot r \cdot (1+r)^n}{(1+r)^n - 1}$, siendo **C** el importe del préstamo (**C4**), **r** el tipo de interés aplicado (**C5**) y **n** el período de amortización (**C6**).

Como ya sabemos, las fórmulas deben empezar con el carácter =, así que el numerador de la expresión anterior se escribirá como

$$C4 * C5/100 * (1 + C5/100) ^ C6$$

mientras el denominador, adoptará la forma

$$(1 + C5/100)^{C6} - 1$$

En definitiva, la celda la celda **F5**, contendrá la expresión


$$C4 * C5/100 * (1 + C5/100) ^ C6 / ((1 + C5/100)^{C6} - 1).$$

La utilización de paréntesis adicionales no afecta a la expresión, siempre que la jerarquía de operaciones no altere el sentido de la fórmula.

La celda **D8** contiene el total de intereses abonados a lo largo de todo el proceso (suma de los contenidos de la columna **D**) mediante la función **SUMA**. En este caso se trata de sumar el contenido de las celdas **D12** hasta **D31**. Además de los dos métodos indicados en la práctica anterior, se puede usar el botón Autosuma, Σ ; conviene seleccionar antes el rango de celdas correspondientes, para evitar los ajustes posteriores.

Para rellenar los años sucesivos (columna **A**) procederemos de la forma siguiente: Introducir un **1** en la celda **A12**, y con el cursor sobre esta celda, elegir **Ser...** en el comando **Rellenar**, menú **Edición**. Las opciones por defecto son las adecuadas para este caso, así que el botón

Aceptar nos lleva directamente a la hoja en la que aparecerán cubiertas las celdas del rango **A13 : A31** con la serie **2, 3, ..., 20**.



La aparición de mensajes de error (división por cero) en alguna celda se debe a la inexistencia de valores iniciales y no debe preocuparnos en este momento.

Aunque es de sobra conocido por todo el alumnado del curso, describimos las operaciones que se deben realizar en la hoja, para los resultados que se desea obtener.

La celda **B12** contiene el valor de la celda **C4**, ya que el capital inicial en el primer año coincide con la cuantía del préstamo solicitado. El resto de las celdas de la columna **B** (capital inicial) corresponden con el valor del capital final del año anterior, si el año correspondiente a cada fila es menor o igual al período de amortización (**C6**).

Las celdas de la columna **D**, intereses abonados, se obtienen aplicando el tipo de interés del préstamo al capital pendiente de amortización al comienzo de cada período.

Las celdas de la columna **C** representan la parte de la anualidad que corresponde al capital amortizado y se obtienen restando de la anualidad (**F5**) los intereses de cada período.

Por último, las celdas de la columna **E**, se calculan restando el capital amortizado del capital inicial.

Describimos, en detalle, el proceso de introducción de los valores de las celdas de las filas **13** y **14** ya que el resto se rellenan con el mismo procedimiento.

En las columnas **B** y **C** aparece un **condicional**, por ejemplo **B13** contiene

SI (A13 < \$C\$6; E12; 0)

cuya lectura sería:

SI el contenido de **A13** es menor que el de **C6**
ENTONCES coloca en esta celda el valor de **E12**
EN OTRO CASO coloca un cero

es decir, **SI** el número de años transcurridos es menor que el número de años de amortización **ENTONCES** el capital inicial (pendiente de amortizar) es el mismo que el capital final del año anterior **EN OTRO CASO** (ha finalizado el período de amortización) el capital pendiente será cero.

Recordar que el símbolo **\$** que aparece delante de las etiquetas de fila y columna (**\$C\$5**, **\$C\$6**, **\$F\$5**), indica referencias absolutas y su función es fijar esta celda como invariable en las operaciones de copia; las referencias relativas (**E12**, **D13**, ...) se ajustan cuando se copian en otra posición de la hoja.

Una vez introducidos los valores en la fila **13** estos serán copiados a todas las demás: Seleccionar el rango **B13: E13** (situar el cursor sobre la celda **B13**, pulsar y, sin soltar el botón del ratón, desplazarlo hasta la celda **E13**; liberar entonces el botón del ratón) y elegir el comando **Copiar** en el menú **Edición**; a continuación, colocar el cursor sobre la celda **B14** y elegir **Pegar** en el mismo menú (o utilizar los botones correspondientes). Las celdas correspondientes de la fila **14** contienen ahora las fórmulas adecuadas para el cálculo.

Para comprobar la corrección de las fórmulas, seleccionar **Fórmulas** en Opciones de ventana, comando **Opciones...** del menú **Herramientas**, y observar la invarianza de las celdas con referencias absolutas y la modificación de las referencias relativas.


	A	B	C	D	E	F
1			CUADRO DE AMORTIZ			
2						
3						
4		Capital				
5		Interés			Anualidad	=C4*C5/100*(1+C5/100)^C6/((1+C5/100)^C6-1)
6		Años				
7						
8		Total intereses	=SUMA(D12:D31)			
9						
10	Año	Capital inicial	Principal	Intereses	Capital final	
11						
12	1	=C4	=F5-D12	=B12*\$C\$5/100	=B12-C12	
13	2	=SI(A13<=\$C\$6;E12;0)	=SI(D13>0;\$F\$5-D13;0)	=B13*\$C\$5/100	=B13-C13	
14	3	=SI(A14<=\$C\$6;E13;0)	=SI(D14>0;\$F\$5-D14;0)	=B14*\$C\$5/100	=B14-C14	
15	4	=SI(A15<=\$C\$6;E14;0)	=SI(D15>0;\$F\$5-D15;0)	=B15*\$C\$5/100	=B15-C15	
16	5	=SI(A16<=\$C\$6;E15;0)	=SI(D16>0;\$F\$5-D16;0)	=B16*\$C\$5/100	=B16-C16	
17	6	=SI(A17<=\$C\$6;E16;0)	=SI(D17>0;\$F\$5-D17;0)	=B17*\$C\$5/100	=B17-C17	
18	7	=SI(A18<=\$C\$6;E17;0)	=SI(D18>0;\$F\$5-D18;0)	=B18*\$C\$5/100	=B18-C18	
19	8	=SI(A19<=\$C\$6;E18;0)	=SI(D19>0;\$F\$5-D19;0)	=B19*\$C\$5/100	=B19-C19	
20	9	=SI(A20<=\$C\$6;E19;0)	=SI(D20>0;\$F\$5-D20;0)	=B20*\$C\$5/100	=B20-C20	
21	10	=SI(A21<=\$C\$6;E20;0)	=SI(D21>0;\$F\$5-D21;0)	=B21*\$C\$5/100	=B21-C21	

Para finalizar, introducimos los datos iniciales correspondientes a un **capital** de 5.000.000 pts., al **interés** del 16 %, durante 10 **años**. Después de unos instantes debe aparecer la hoja cubierta con los valores calculados.

	A	B	C	D	E	F
1			CUADRO DE AMORTIZACION DE UN PRESTAMO			
2						
3						
4		Capital	5.000.000 Pts			
5		Interés	16,00		Anualidad ..	1.034.505,42 Pts
6		Años	10			
7						
8		Total intereses	5.345.054,15 Pts			
9						
10	Año	Capital inicial	Principal	Intereses	Capital final	
11						
12	1	5.000.000,00	234.505,42	800.000,00	4.765.494,58	
13	2	4.765.494,58	272.026,28	762.479,13	4.493.468,30	
14	3	4.493.468,30	315.550,49	718.954,93	4.177.917,82	
15	4	4.177.917,82	366.038,56	668.466,85	3.811.879,25	
16	5	3.811.879,25	424.604,74	609.900,68	3.387.274,52	
17	6	3.387.274,52	492.541,49	541.963,92	2.894.733,02	
18	7	2.894.733,02	571.348,13	463.157,28	2.323.384,89	
19	8	2.323.384,89	662.763,83	371.741,58	1.660.621,06	
20	9	1.660.621,06	768.806,05	265.699,37	891.815,01	
21	10	891.815,01	891.815,01	142.690,40	0,00	

ACTIVIDADES EXCEL #2


- 1.- Calcular la anualidad y el cuadro de amortización de un préstamo de 10.000.000 de ptas. a pagar en 13 años al 16% de interés. (Este ejercicio puede servir para estudiar la variación de los intereses al aumentar el importe del préstamo y/o el número de años). Nombrar la hoja como Anualidad1 y guardar el libro con el nombre **Anualidades.xls**.
- 2.- En el mismo libro que la actividad anterior, cambiar a la Hoja2 y nombrarla como Anualidad2. En esta, se deben rellenar las celdas adecuadas para calcular el capital que se puede solicitar a una entidad financiera que aplica un tipo de interés del 12% anual si estamos dispuestos a pagar una anualidad de 1.000.000 de ptas. durante 8 años. Guardar de nuevo el libro como **Anualidades.xls**.
- 3.- Extender el modelo ampliando el plazo de amortización posible hasta 30 años, en una nueva hoja de nombre ampliacion. Guardar de nuevo el libro con el mismo nombre **Anualidades.xls**.
- 4.- Con el mismo libro, crea una nueva hoja de nombre trimestral para calcular la anualidad y el cuadro de amortización del préstamo descrito en el ejercicio 1, suponiendo que los pagos se abonan


trimestralmente (el número de plazos de amortización serán ahora 52 y el tipo de interés a aplicar del 4%). Guardar de nuevo el libro con el mismo nombre **Anualidades.xls** y **envialo al tutor** .

GUIA ACTIVIDADES EXCEL #2


La única particularidad de las actividades que puede resultar desconocida es la relativa a la creación, cambio de nombre, reordenación y eliminación de hojas de un libro. Explicaremos estos procedimientos en detalle.

Recordamos que en la parte inferior del libro se muestran, por defecto, tres separadores con los nombres **Hoja1**, **Hoja2** y **Hoja3**.




Para cambiar el orden, no hay más que pulsar sobre el indicador de una de las hojas y, sin soltar el botón izquierdo del ratón (el cursor adopta la forma del botón Nuevo ), desplazar la hoja a la nueva posición.


La pulsación del botón derecho del ratón, cuando el cursor se encuentra sobre uno de los nombres de hoja, conduce a un menú desde el que podemos añadir, borrar, renombrar y mover o copiar hojas.



La opción **I**ntertar... abre una ventana en la que se puede seleccionar el elemento a añadir (hojas y gráficos son los más habituales). Las opciones **E**liminar y **C**ambiar nombre no requieren explicaciones adicionales, hacen lo que su nombre indica.



Mover o copiar... abre una ventana que permite seleccionar el libro de destino y la posición de la nueva hoja.



La opción Seleccionar todas las hojas, se puede utilizar para crear copias en las hojas seleccionadas de los datos o fórmulas introducidos en cualquiera de ellas.